

# AVL Trees

D Oisín Kidney

July 28, 2018

## Abstract

This is a verified implementation of AVL trees in Agda, taking ideas primarily from Conor McBride’s paper “How to Keep Your Neighbours in Order” [2] and the Agda standard library [1].

## Contents

<b>1</b>	<b>Introduction</b>	<b>1</b>
<b>2</b>	<b>Bounded</b>	<b>2</b>
<b>3</b>	<b>Balance</b>	<b>3</b>
<b>4</b>	<b>The Tree Type</b>	<b>3</b>
<b>5</b>	<b>Rotations</b>	<b>4</b>
5.1	Right Rotation . . . . .	4
5.2	Left Rotation . . . . .	5
<b>6</b>	<b>Insertion</b>	<b>6</b>
<b>7</b>	<b>Lookup</b>	<b>6</b>
<b>8</b>	<b>Deletion</b>	<b>7</b>
8.1	Uncons . . . . .	7
8.2	Widening . . . . .	8
8.3	Full Deletion . . . . .	9
<b>9</b>	<b>Packaging</b>	<b>10</b>
9.1	Dependent Map . . . . .	10
9.2	Non-Dependent (Simple) Map . . . . .	10
9.3	Set . . . . .	11

## 1 Introduction

First, some imports.

```

{-# OPTIONS --without-K #-}

open import Relation.Binary
open import Relation.Binary.PropositionalEquality
open import Level using (Lift; lift;  $\perp$ _; lower)
open import Data.Nat as  $\mathbb{N}$  using ( $\mathbb{N}$ ; suc; zero; pred)
open import Data.Product
open import Data.Unit renaming (T to 1)
open import Data.Maybe
open import Function
open import Data.Bool
open import Data.Empty renaming ( $\perp$  to 0)

```

Next, we declare a module: the entirety of the following code is parameterized over the *key* type, and a strict total order on that key.

```

module AVL
  {k r} (Key : Set k)
  {_ <_ : Rel Key r}
  (isStrictTotalOrder : IsStrictTotalOrder  $\equiv$  _ <_)
  where

  open IsStrictTotalOrder isStrictTotalOrder

```

## 2 Bounded

The basic idea of the verified implementation is to store in each leaf a proof that the upper and lower bounds of the trees to its left and right are ordered appropriately.

Accordingly, the tree type itself will have to have the upper and lower bounds in its indices. But what are the upper and lower bounds of a tree with no neighbours? To describe this case, we add lower and upper bounds to our key type.

```

module Bounded where

infix 5  $\lfloor$   $\rfloor$ 

data  $\mathbb{I}$  : Set k where
   $\perp$   $\top$  :  $\mathbb{I}$ 
   $\lfloor$   $\rfloor$  : (k : Key)  $\rightarrow$   $\mathbb{I}$ 

```

This type itself admits an ordering relation.

```

infix 4  $\mathbb{I}$  <_

 $\mathbb{I}$  <_ :  $\mathbb{I} \rightarrow \mathbb{I} \rightarrow$  Set r
 $\perp$  <_  $\perp$  = Lift r 0
 $\perp$  <_  $\top$  = Lift r 1

```

$$\begin{aligned}
\perp & \quad \top < [\_] = \text{Lift } r \ 1 \\
\top & \quad \perp < \_ = \text{Lift } r \ 0 \\
[\_] & \quad \perp < \perp = \text{Lift } r \ 0 \\
[\_] & \quad \perp < \top = \text{Lift } r \ 1 \\
[x] & \quad \perp < [y] = x < y
\end{aligned}$$

Finally, we can describe a value as being “in bounds” like so.

$$\begin{aligned}
& \text{infix 4 } \_ < \_ < \_ \\
& \_ < \_ < \_ : \top \rightarrow \text{Key} \rightarrow \top \rightarrow \text{Set } r \\
& l < x < u = l \top < [x] \times [x] \top < u
\end{aligned}$$

### 3 Balance

To describe the balance of the tree, we use the following type:

$$\begin{aligned}
& \text{data } \langle \_ \sqcup \_ \rangle \equiv \_ : \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N} \rightarrow \text{Set where} \\
& \swarrow : \forall \{n\} \rightarrow \langle \text{suc } n \sqcup \_ \rangle \equiv \text{suc } n \\
& \dashv : \forall \{n\} \rightarrow \langle \_ \sqcup n \rangle \equiv n \\
& \searrow : \forall \{n\} \rightarrow \langle \_ \sqcup \text{suc } n \rangle \equiv \text{suc } n
\end{aligned}$$

The tree can be either left- or right-heavy (by one), or even. The indices of the type are phrased as a proof:

$$\max(x, y) = z \tag{1}$$

The height of a tree is the maximum height of its two subtrees, plus one. Storing a proof of the maximum in this way will prove useful later.

We will also need some combinators for balance:

$$\begin{aligned}
& \vdash : \forall \{x \ y \ z\} \rightarrow \langle x \sqcup y \rangle \equiv z \rightarrow \langle z \sqcup x \rangle \equiv z \\
& \vdash \swarrow = \dashv \\
& \vdash \dashv = \dashv \\
& \vdash \searrow = \swarrow \\
& \vdash : \forall \{x \ y \ z\} \rightarrow \langle x \sqcup y \rangle \equiv z \rightarrow \langle y \sqcup z \rangle \equiv z \\
& \vdash \swarrow = \searrow \\
& \vdash \dashv = \dashv \\
& \vdash \searrow = \dashv
\end{aligned}$$

### 4 The Tree Type

The type itself is indexed by the lower and upper bounds, some value to store with the keys, and a height. In using the balance type defined earlier, we ensure that the children of a node cannot differ in height by more than 1. The bounds proofs also ensure that the tree must be ordered correctly.

```

data Tree {v} (V : Key → Set v) (l u :  $\mathbb{T}$ ) :  $\mathbb{N}$  → Set (k  $\sqcup$  v  $\sqcup$  r) where
leaf   : (l < u : l  $\mathbb{T}$  < u) → Tree V l u 0
node   :  $\forall \{h \text{ lh rh}\}$ 
        (k : Key)
        (v : V k)
        (bl : (lh  $\sqcup$  rh)  $\equiv$  h)
        (lk : Tree V l [k] lh)
        (ku : Tree V [k] u rh) →
        Tree V l u (suc h)

```

## 5 Rotations

AVL trees are rebalanced by rotations: if, after an insert or deletion, the balance invariant has been violated, one of these rotations is performed as correction.

Before we implement the rotations, we need a type to describe a tree whose height may have changed:

```

Inserted :  $\forall \{v\} (V : Key \rightarrow Set v) (l u : \mathbb{T}) (n : \mathbb{N}) \rightarrow Set (k \sqcup v \sqcup r)$ 
Inserted V l u n =  $\exists [inc?] Tree V l u (if inc? then suc n else n)$ 
pattern 0+ tr = false , tr
pattern 1+ tr = true  , tr

```

### 5.1 Right Rotation

When the left subtree becomes too heavy, we rotate the tree to the right.

```

rotr :  $\forall \{lb \text{ ub rh } v\} \{V : Key \rightarrow Set v\}$ 
        → (k : Key)
        → V k
        → Tree V lb [k] (suc (suc rh))
        → Tree V [k] ub rh
        → Inserted V lb ub (suc (suc rh))

```

This rotation comes in two varieties: single and double. Single rotation can be seen in figure 1.




Figure 1: Single right-rotation

```

rotr x xv (node y yv  $\swarrow$  a b) c = 0+ (node y yv  $\swarrow$  a (node x xv  $\swarrow$  b c))
rotr x xv (node y yv  $\swarrow$  a b) c = 1+ (node y yv  $\swarrow$  a (node x xv  $\swarrow$  b c))

```




Figure 2: Double right-rotation

And double rotation in figure 2.

$$\begin{aligned} \text{rot}^r \ x \ x v \ (\text{node } y \ y v \searrow a \ (\text{node } z \ z v \text{ bl } b \ c)) \ d = \\ 0+ \ (\text{node } z \ z v \dashv (\text{node } y \ y v \dashv \text{ bl } a \ b) \ (\text{node } x \ x v \dashv \text{ bl } c \ d)) \end{aligned}$$

## 5.2 Left Rotation

Left-rotation is essentially the inverse of right.

$$\begin{aligned} \text{rot}^l : \forall \{lb \ ub \ lh \ v\} \{V : \text{Key} \rightarrow \text{Set } v\} \\ \rightarrow (k : \text{Key}) \\ \rightarrow V \ k \\ \rightarrow \text{Tree } V \ lb \ [k] \ lh \\ \rightarrow \text{Tree } V \ [k] \ ub \ (\text{suc } (\text{suc } lh)) \\ \rightarrow \text{Inserted } V \ lb \ ub \ (\text{suc } (\text{suc } lh)) \end{aligned}$$

Single (seen in figure 3).




Figure 3: Single left-rotation

$$\begin{aligned} \text{rot}^l \ x \ x v \ c \ (\text{node } y \ y v \searrow b \ a) &= 0+ \ (\text{node } y \ y v \dashv (\text{node } x \ x v \dashv c \ b) \ a) \\ \text{rot}^l \ x \ x v \ c \ (\text{node } y \ y v \dashv b \ a) &= 1+ \ (\text{node } y \ y v \searrow (\text{node } x \ x v \searrow c \ b) \ a) \end{aligned}$$

and double (figure 4):

$$\begin{aligned} \text{rot}^l \ x \ x v \ d \ (\text{node } y \ y v \searrow (\text{node } z \ z v \text{ bl } c \ b) \ a) = \\ 0+ \ (\text{node } z \ z v \dashv (\text{node } x \ x v \dashv \text{ bl } d \ c) \ (\text{node } y \ y v \dashv \text{ bl } b \ a)) \end{aligned}$$




Figure 4: Double left-rotation

## 6 Insertion

After the rotations, insertion is relatively easy. We allow the caller to supply a combining function.

```

insert : ∀ {l u h v} {V : Key → Set v} (k : Key)
  → V k
  → (V k → V k → V k)
  → Tree V l u h
  → l < k < u
  → Inserted V l u h
insert v vc f (leaf l < u) (l, u) = 1+ (node v vc - (leaf l) (leaf u))
insert v vc f (node k kc bl tl tr) prf with compare v k
insert v vc f (node k kc bl tl tr) (l, _)
  | tri < a _ _ with insert v vc f tl (l, a)
... | 0+ tl' = 0+ (node k kc bl tl' tr)
... | 1+ tl' with bl
...   | / = rotr k kc tl' tr
...   | - = 1+ (node k kc / tl' tr)
...   | \ = 0+ (node k kc - tl' tr)
insert v vc f (node k kc bl tl tr) _
  | tri ≈ _ refl _ = 0+ (node k (f vc kc) bl tl tr)
insert v vc f (node k kc bl tl tr) (_, u)
  | tri > _ _ c with insert v vc f tr (c, u)
... | 0+ tr' = 0+ (node k kc bl tl tr')
... | 1+ tr' with bl
...   | / = 0+ (node k kc - tl tr')
...   | - = 1+ (node k kc \ tl tr')
...   | \ = rotl k kc tl tr'

```

## 7 Lookup

Lookup is also very simple. No invariants are needed here.

```

lookup : (k : Key)
  → ∀ {l u s v} {V : Key → Set v}

```

```

→ Tree V l u s
→ Maybe (V k)
lookup k (leaf l < u) = nothing
lookup k (node v vc _ tl tr) with compare k v
... | tri < _ _ _ = lookup k tl
... | tri ≈ _ refl _ = just vc
... | tri > _ _ _ = lookup k tr

```

## 8 Deletion

Deletion is by far the most complex operation out of the three provided here. For deletion from a normal BST, you go to the node where the desired value is, perform an “uncons” operation on the right subtree, and use that to rebuild and rebalance the tree.

### 8.1 Uncons

First then, we need to define “uncons”. We’ll use a custom type as the return type from our uncons function, which stores the minimum element from the tree, and the rest of the tree:

```

data Deleted {v} (V : Key → Set v) (lb ub :  $\mathbb{T}$ ) :  $\mathbb{N}$  → Set (k  $\sqcup$  v  $\sqcup$  r) where
  -0 :  $\forall \{n\} \rightarrow$  Tree V lb ub n → Deleted V lb ub n
  -1 :  $\forall \{n\} \rightarrow$  Tree V lb ub n → Deleted V lb ub (suc n)

deleted :  $\forall \{v\} \{V : Key \rightarrow Set v\} \{lb ub n\}$ 
  → Inserted V lb ub n
  → Deleted V lb ub (suc n)
deleted (0+ x) = x -1
deleted (1+ x) = x -0

record Cons {v}
  (V : Key → Set v)
  (lb ub :  $\mathbb{T}$ )
  (h :  $\mathbb{N}$ ) : Set (k  $\sqcup$  v  $\sqcup$  r) where
  constructor cons
  field
    head : Key
    val : V head
    l < u : lb  $\mathbb{T}$  < [ head ]
    tail : Deleted V [ head ] ub h

```

You’ll notice it also stores a proof that the extracted element preserves the lower bound.

The uncons function itself is written in a continuation-passing style.

```

uncons :  $\forall \{lb ub h lh rh v\} \{V : Key \rightarrow Set v\}$ 
  → (k : Key)
  → V k

```

```

→ ⟨ lh ⊔ rh ⟩≡ h
→ Tree V lb [ k ] lh
→ Tree V [ k ] ub rh
→ Cons V lb ub (suc h)
uncons k v bl tl tr = go k v bl tl tr id
where
go : ∀ {lb ub h lh rh v ub' h'} {V : Key → Set v}
→ (k : Key)
→ V k
→ ⟨ lh ⊔ rh ⟩≡ h
→ Tree V lb [ k ] lh
→ Tree V [ k ] ub rh
→ (∀ {lb'} → Deleted V [ lb' ] ub (suc h) → Deleted V [ lb' ] ub' (suc h'))
→ Cons V lb ub' (suc h')
go k v ⊖ (leaf l<u) tr c = cons k v l<u (c (tr -1))
go k v ⊖ (node kl vl bll tll trl) tr c = go kl vl bll tll trl
λ { (tl' -1) → c ((node k v ⊖ tl' tr) -0)
; (tl' -0) → c ((node k v ⊖ tl' tr) -0) }
go k v ⊗ (leaf l<u) tr c = cons k v l<u (c (tr -1))
go k v ⊗ (node kl vl bll tll trl) tr c = go kl vl bll tll trl
λ { (tl' -1) → c (deleted (rotl k v tl' tr))
; (tl' -0) → c ((node k v ⊗ tl' tr) -0) }
go k v ⊘ (node kl vl bll tll trl) tr c = go kl vl bll tll trl
λ { (tl' -1) → c ((node k v ⊖ tl' tr) -1)
; (tl' -0) → c ((node k v ⊘ tl' tr) -0) }

```

## 8.2 Widening

To join the two subtrees together after a deletion operation, we need to weaken (or widen) the bounds of the left tree. This is an  $\mathcal{O}(\log n)$  operation.

For the widening, we'll need some properties on orderings:

```

x⋈⊥ : ∀ {x} → x ⊔ ⊥ < ⊥ → Lift r 0
x⋈⊥ {⊥} = lift ∘ lower
x⋈⊥ {⊤} = lift ∘ lower
x⋈⊥ {[ _ ]} = lift ∘ lower

⊔<-trans : ∀ x {y z} → x ⊔ < y → y ⊔ < z → x ⊔ < z
⊔<-trans ⊥ {y} {⊥} _ y<z = x⋈⊥ {x = y} y<z
⊔<-trans ⊥ {_} {⊤} _ _ = _
⊔<-trans ⊥ {[ _ ]} {[ _ ]} _ _ = _
⊔<-trans ⊤ {[ _ ]} {[ _ ]} (lift ()) _
⊔<-trans [ _ ] {y} {⊥} _ y<z = x⋈⊥ {x = y} y<z
⊔<-trans [ _ ] {[ _ ]} {⊤} _ _ = _
⊔<-trans [ _ ] {[ _ ]} {[ _ ]} (lift ()) _

```



```

 $\top$ <-trans [ _ ] { $\top$ } { [ _ ] } _ (lift ())
 $\top$ <-trans [ x ] { [ y ] } { [ z ] } x<y y<z =
  IsStrictTotalOrder.trans isStrictTotalOrder x<y y<z

```

Finally, the widen function itself simply walks down the right branch of the tree until it hits a leaf.

```

widen :  $\forall \{lb\ ub\ ub'\ h\ v\} \{V : Key \rightarrow Set\ v\}$ 
   $\rightarrow ub \top < ub'$ 
   $\rightarrow Tree\ V\ lb\ ub\ h$ 
   $\rightarrow Tree\ V\ lb\ ub'\ h$ 
widen {lb} ub<ub' (leaf l<u) = leaf ( $\top$ <-trans lb l<u ub<ub')
widen ub<ub' (node k v bl tl tr) = node k v bl tl (widen ub<ub' tr)

```

### 8.3 Full Deletion

The deletion function is by no means simple, but it does maintain the correct complexity bounds.

```

delete :  $\forall \{lb\ ub\ h\ v\} \{V : Key \rightarrow Set\ v\}$ 
   $\rightarrow (k : Key)$ 
   $\rightarrow Tree\ V\ lb\ ub\ h$ 
   $\rightarrow Deleted\ V\ lb\ ub\ h$ 
delete _ (leaf l<u) = leaf l<u -0
delete k (node k1 v b tl tr) with compare k k1
delete k (node k1 v b tl tr) | tri< _ _ _ with delete k tl | b
... | tl' -1 |  $\swarrow$  = node k1 v  $\neg$  tl' tr -1
... | tl' -1 |  $\neg$  = node k1 v  $\searrow$  tl' tr -0
... | tl' -1 |  $\searrow$  = deleted (rotl k1 v tl' tr)
... | tl' -0 | _ = node k1 v b tl' tr -0
delete {lb} k (node k v b tl (leaf k<ub)) | tri $\approx$  _ refl _ with b | tl
... |  $\swarrow$  | _ = widen k<ub tl -1
... |  $\neg$  | leaf lb<k = leaf ( $\top$ <-trans lb lb<k k<ub) -1
delete k (node k v b tl (node kr vr br tlr trr)) | tri $\approx$  _ refl _
  with b | uncons kr vr br tlr trr
... |  $\swarrow$  | cons k' v' l<u (tr' -1) = deleted (rotr k' v' (widen l<u tl) tr')
... |  $\swarrow$  | cons k' v' l<u (tr' -0) = node k' v'  $\swarrow$  (widen l<u tl) tr' -0
... |  $\neg$  | cons k' v' l<u (tr' -1) = node k' v'  $\swarrow$  (widen l<u tl) tr' -0
... |  $\neg$  | cons k' v' l<u (tr' -0) = node k' v'  $\neg$  (widen l<u tl) tr' -0
... |  $\searrow$  | cons k' v' l<u (tr' -1) = node k' v'  $\neg$  (widen l<u tl) tr' -1
... |  $\searrow$  | cons k' v' l<u (tr' -0) = node k' v'  $\searrow$  (widen l<u tl) tr' -0
delete k (node k1 v b tl tr) | tri> _ _ _ with delete k tr | b
... | tr' -1 |  $\swarrow$  = deleted (rotr k1 v tl tr')
... | tr' -1 |  $\neg$  = node k1 v  $\swarrow$  tl tr' -0
... | tr' -1 |  $\searrow$  = node k1 v  $\neg$  tl tr' -1
... | tr' -0 | _ = node k1 v b tl tr' -0

```

## 9 Packaging

Users don't need to be exposed to the indices on the full tree type: here, we package it in three forms.

### 9.1 Dependent Map

```

module DependantMap where
data Map {v} (V : Key → Set v) : Set (k ⊔ v ⊔ r) where
  tree : ∀ {h} → Bounded.Tree V Bounded.⊥ Bounded.T h → Map V

insertWith : ∀ {v} {V : Key → Set v} (k : Key)
  → V k
  → (V k → V k → V k)
  → Map V
  → Map V
insertWith k v f (tree tr) =
  tree (proj₂ (Bounded.insert k v f tr (lift tt , lift tt)))

insert : ∀ {v} {V : Key → Set v} (k : Key) → V k → Map V → Map V
insert k v = insertWith k v const

lookup : (k : Key) → ∀ {v} {V : Key → Set v} → Map V → Maybe (V k)
lookup k (tree tr) = Bounded.lookup k tr

delete : (k : Key) → ∀ {v} {V : Key → Set v} → Map V → Map V
delete k (tree tr) with Bounded.delete k tr
... | tr' Bounded.-0 = tree tr'
... | tr' Bounded.-1 = tree tr'

```

### 9.2 Non-Dependent (Simple) Map

```

module Map where
data Map {v} (V : Set v) : Set (k ⊔ v ⊔ r) where
  tree : ∀ {h} → Bounded.Tree (const V) Bounded.⊥ Bounded.T h → Map V

insertWith : ∀ {v} {V : Set v} (k : Key) → V → (V → V → V) → Map V → Map V
insertWith k v f (tree tr) =
  tree (proj₂ (Bounded.insert k v f tr (lift tt , lift tt)))

insert : ∀ {v} {V : Set v} (k : Key) → V → Map V → Map V
insert k v = insertWith k v const

lookup : (k : Key) → ∀ {v} {V : Set v} → Map V → Maybe V
lookup k (tree tr) = Bounded.lookup k tr

delete : (k : Key) → ∀ {v} {V : Set v} → Map V → Map V
delete k (tree tr) with Bounded.delete k tr

```

```

... | tr' Bounded.-0 = tree tr'
... | tr' Bounded.-1 = tree tr'

```

### 9.3 Set

Note that we can't call the type itself “Set”, as that's a reserved word in Agda.

```

module Sets where
data ⟨Set⟩ : Set (k ⊔ r) where
  tree : ∀ {h} → Bounded.Tree (const 1) Bounded.⊥ Bounded.⊤ h → ⟨Set⟩

insert : Key → ⟨Set⟩ → ⟨Set⟩
insert k (tree tr) =
  tree (proj₂ (Bounded.insert k tt const tr (lift tt , lift tt)))

member : Key → ⟨Set⟩ → Bool
member k (tree tr) = is-just (Bounded.lookup k tr)

delete : (k : Key) → ⟨Set⟩ → ⟨Set⟩
delete k (tree tr) with Bounded.delete k tr
... | tr' Bounded.-0 = tree tr'
... | tr' Bounded.-1 = tree tr'

```

## References

- [1] N. A. Danielsson, “The Agda standard library.” [Online]. Available: <https://agda.github.io/agda-stdlib/README.html>
- [2] C. T. McBride, “How to Keep Your Neighbours in Order,” in *Proceedings of the 19th ACM SIG-PLAN International Conference on Functional Programming*, ser. ICFP '14. ACM, pp. 297–309. [Online]. Available: <https://personal.cis.strath.ac.uk/conor.mcbride/pub/Pivotal.pdf>